
12 Policy Gradient Optimization

We can use estimates of the policy gradient to drive the search of the parameter
space toward an optimal policy. The previous chapter outlined methods for
estimating this gradient. This chapter explains how to use these estimates to
guide the optimization. We begin with gradient ascent, which simply takes steps
in the direction of the gradient at each iteration. Determining the step size is a
major challenge. Large steps can lead to faster progress to the optimum, but they
can overshoot. The natural policy gradient modifies the direction of the gradient
to better handle variable levels of sensitivity across parameter components. We
conclude with the trust region method, which starts in exactly the same way as
the natural gradient method to obtain a candidate policy. It then searches along
the line segment in policy space connecting the original policy to this candidate
to find a better policy.

12.1 Gradient Ascent Update

We can use gradient ascent (reviewed in appendix A.11) to find a policy parame-
terized by θ that maximizes the expected utility U(θ). Gradient ascent is a type
of iterated ascent method, which involves taking steps in the parameter space at
each iteration in an attempt to improve the quality of the associated policy. All
the methods discussed in this chapter are iterated ascent methods, but they differ
in how they take steps. The gradient ascent method discussed in this section
takes steps in the direction of ∇U(θ), which may be estimated using one of the
methods discussed in the previous chapter. The update of θ is

θ← θ+ α∇U(θ) (12.1)

where the step length is equal to a step factor α > 0 times the magnitude of the
gradient.

Algorithm 12.1 implements a method that takes such a step. This method can
be called for either a fixed number of iterations or until θ or U(θ) converges.
Gradient ascent, as well as the other algorithms discussed in this chapter, is not
guaranteed to converge to the optimal policy. However, there are techniques to
encourage convergence to a locally optimal policy, inwhich taking an infinitesimally
small step in parameter space cannot result in a better policy. One approach is to
decay the step factor with each step.1

1 This approach, as well as many
others, are covered in detail byM. J.
Kochenderfer and T.A. Wheeler,
Algorithms for Optimization. MIT
Press, 2019.

struct PolicyGradientUpdate
∇U # policy gradient estimate
α # step factor

end

function update(M::PolicyGradientUpdate, θ)
return θ + M.α * M.∇U(θ)

end

Algorithm 12.1. The gradient as-
cent method for policy optimiza-
tion. It takes a step from a point
θ in the direction of the gradient
∇U with step factor α. We can use
one of the methods in the previous
chapter to compute ∇U.

Very large gradients tend to overshoot the optimum and may occur due to a
variety of reasons. Rewards for some problems, such as for the 2048 problem
(appendix F.2), can vary by orders of magnitude. One approach for keeping the
gradients manageable is to use gradient scaling, which limits the magnitude of a
gradient estimate before using it to update the policy parameterization. Gradients
are commonly limited to having an L2-norm of 1. Another approach is gradient
clipping, which conducts elementwise clamping of the gradient before using it to
update the policy. Clipping commonly limits the entries to lie between ±1. Both
techniques are implemented in algorithm 12.2.

scale_gradient(∇, L2_max) = min(L2_max/norm(∇), 1)*∇
clip_gradient(∇, a, b) = clamp.(∇, a, b)

Algorithm 12.2. Methods for gra-
dient scaling and clipping. Gradi-
ent scaling limits the magnitude
of the provided gradient vector ∇
to L2_max. Gradient clipping pro-
vides elementwise clamping of the
provided gradient vector ∇ to be-
tween a and b.

Scaling and clipping differ in how they affect the final gradient direction, as
demonstrated in figure 12.1. Scaling will leave the direction unaffected, whereas
clipping affects each component individually. Whether this difference is advanta-
geous depends on the problem. For example, if a single component dominates
the gradient vector, scaling will zero out the other components.

Figure 12.1. The effect of gradi-
ent scaling and clipping applied to
the simple regulator problem. Each
gradient evaluation ran 10 rollouts
to depth 10. Step updates were ap-
plied with a step size of 0.2. The
optimal policy parameterization is
shown in black.

12.2 Restricted Gradient Update

The remaining algorithms in this chapter attempt to optimize an approximation
of the objective function U(θ), subject to a constraint that the policy parameters at
the next step θ′ are not too far from θ at the current step. The constraint takes the
form g(θ,θ′) ≤ ǫ, where ǫ > 0 is a free parameter in the algorithm. The methods
differ in their approximation of U(θ) and the form of g. This section describes a
simple restricted step method.

We use the first-order Taylor approximation (appendix A.12) obtained from
our gradient estimate at θ to approximate U:

U(θ′) ≈ U(θ) +∇U(θ)⊤(θ′ − θ) (12.2)

For the constraint, we use

g(θ,θ′) =
1

2
(θ′ − θ)⊤I(θ′ − θ) = 1

2
‖θ′ − θ‖2

2 (12.3)

We can view this constraint as limiting the step length to no more than
√

2ǫ.
In other words, the feasible region in our optimization is a ball of radius

√
2ǫ

centered at θ.
The optimization problem is, then,

maximize
θ′

U(θ) +∇U(θ)⊤(θ′ − θ)

subject to
1

2
(θ′ − θ)⊤I(θ′ − θ) ≤ ǫ

(12.4)

We can drop U(θ) from the objective since it does not depend on θ′. In addition,
we can change the inequality to an equality in the constraint because the linear
objective forces the optimal solution to be on the boundary of the feasible region.
These changes result in an equivalent optimization problem:

maximize
θ′

∇U(θ)⊤(θ′ − θ)

subject to
1

2
(θ′ − θ)⊤I(θ′ − θ) = ǫ

(12.5)

This optimization problem can be solved analytically:

θ′ = θ+ u

√

2ǫ

u⊤u
= θ+

√
2ǫ

u

‖u‖ (12.6)

where the unnormalized search direction u is simply ∇U(θ). Of course, we do
not know ∇U(θ) exactly, but we can use any of the methods described in the
previous chapter to estimate it. Algorithm 12.3 provides an implementation.

struct RestrictedPolicyUpdate
𝒫 # problem
b # initial state distribution
d # depth
m # number of samples
∇logπ # gradient of log likelihood
π # policy
ϵ # divergence bound

end

function update(M::RestrictedPolicyUpdate, θ)
𝒫, b, d, m, ∇logπ, π, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.π, M.𝒫.γ
πθ(s) = π(θ, s)
R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))
τs = [simulate(𝒫, rand(b), πθ, d) for i in 1:m]
∇log(τ) = sum(∇logπ(θ, a, s) for (s,a) in τ)
∇U(τ) = ∇log(τ)*R(τ)
u = mean(∇U(τ) for τ in τs)
return θ + u*sqrt(2*M.ϵ/dot(u,u))

end

Algorithm 12.3. The update func-
tion for the restricted policy gra-
dient method at θ for a problem
𝒫 with initial state distribution b.
The gradient is estimated from an
initial state distribution b to depth
d with m simulations of parameter-
ized policy π(θ, s)with log policy
gradient ∇logπ.

12.3 Natural Gradient Update

The natural gradientmethod2 is a variation of the restricted step method discussed 2 S. Amari, “Natural Gradient
Works Efficiently in Learning,”
Neural Computation, vol. 10, no. 2,
pp. 251–276, 1998.

in the previous section to better handle situations when some components of the
parameter space are more sensitive than others. Sensitivity in this context refers
to how much the utility of a policy varies with respect to small changes in one of
the parameters. The sensitivity in gradient methods is largely determined by the
choice of scaling of the policy parameters. The natural policy gradient method
makes the search direction u invariant to parameter scaling. Figure 12.2 illustrates
the differences between the true gradient and the natural gradient.

−2 −1 0

0

0.2

0.4

θ1

θ 2

true gradient

−2 −1 0

θ1

natural gradient Figure 12.2. A comparison of
the true gradient and the natu-
ral gradient on the simple regu-
lator problem (see appendix F.5).
The true gradient generally points
strongly in the negative θ2 direc-
tion, whereas the natural gradient
generally points toward the opti-
mum (black dot) at [−1, 0]. A sim-
ilar figure is presented in J. Pe-
ters and S. Schaal, “Reinforcement
Learning of Motor Skills with Pol-
icy Gradients,” Neural Networks,
vol. 21, no. 4, pp. 682–697, 2008.

The natural policy gradient method uses the same first-order approximation
of the objective as in the previous section. The constraint, however, is different.
The intuition is that we want to restrict changes in θ that result in large changes
in the distribution over trajectories. A way to measure how much a distribution
changes is to use the Kullback-Leibler divergence, or KL divergence (appendix A.10).
We could impose the constraint

g(θ,θ′) = DKL
(

p(· | θ)
∣

∣

∣

∣ p(· | θ′)
)

≤ ǫ (12.7)

but instead we will use a second-order Taylor approximation:

g(θ,θ′) =
1

2
(θ′ − θ)⊤Fθ(θ

′ − θ) ≤ ǫ (12.8)

where the Fisher information matrix has the following form:

Fθ =
∫

p(τ | θ)∇ log p(τ | θ)∇ log p(τ | θ)⊤ dτ (12.9)

= Eτ

[

∇ log p(τ | θ)∇ log p(τ | θ)⊤
]

(12.10)

The resulting optimization problem is

maximize
θ′

∇U(θ)⊤(θ′ − θ)

subject to
1

2
(θ′ − θ)⊤Fθ(θ

′ − θ) = ǫ
(12.11)

which looks identical to equation (12.5) except that instead of the identity matrix
I, we have the Fisher matrix Fθ. This difference results in an ellipsoid feasible set.
Figure 12.3 shows an example in two dimensions.

∇U(θ)

F−1∇U(θ)

1
2 ∆θ⊤Fθ∆θ = ǫ

Figure 12.3. The natural policy gra-
dient places a constraint on the
approximated Kullback-Leibler di-
vergence. This constraint takes the
form of an ellipse. The ellipse may
be elongated in certain directions,
allowing larger steps if the gradi-
ent is rotated.

This optimization problem can be solved analytically and has the same form
as the update in the previous section:

θ′ = θ+ u

√

2ǫ

∇U(θ)⊤u
(12.12)

except that we now have3

3 This computation can be done
using conjugate gradient descent,
which reduces computation when
the dimension of θ is large. S.M.
Kakade, “A Natural Policy Gradi-
ent,” in Advances in Neural Infor-
mation Processing Systems (NIPS),
2001.

u = F−1
θ
∇U(θ) (12.13)

We can use sampled trajectories to estimate Fθ and ∇U(θ). Algorithm 12.4 pro-
vides an implementation.

12.4 Trust Region Update

This section discusses a method for searching within the trust region, defined by
the elliptical feasible region from the previous section. This category of approach
is referred to as trust region policy optimization (TRPO).4 It works by computing 4 J. Schulman, S. Levine, P. Moritz,

M. Jordan, and P. Abbeel, “Trust
Region Policy Optimization,” in In-
ternational Conference on Machine
Learning (ICML), 2015.

the next evaluation point θ′ that would be taken by the natural policy gradient
and then conducting a line search along the line segment connecting θ to θ′. A key
property of this line search phase is that evaluations of the approximate objective
and constraint do not require any additional rollout simulations.

struct NaturalPolicyUpdate
𝒫 # problem
b # initial state distribution
d # depth
m # number of samples
∇logπ # gradient of log likelihood
π # policy
ϵ # divergence bound

end

function natural_update(θ, ∇f, F, ϵ, τs)
∇fθ = mean(∇f(τ) for τ in τs)
u = mean(F(τ) for τ in τs) \ ∇fθ
return θ + u*sqrt(2ϵ/dot(∇fθ,u))

end

function update(M::NaturalPolicyUpdate, θ)
𝒫, b, d, m, ∇logπ, π, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.π, M.𝒫.γ
πθ(s) = π(θ, s)
R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))
∇log(τ) = sum(∇logπ(θ, a, s) for (s,a) in τ)
∇U(τ) = ∇log(τ)*R(τ)
F(τ) = ∇log(τ)*∇log(τ)'
τs = [simulate(𝒫, rand(b), πθ, d) for i in 1:m]
return natural_update(θ, ∇U, F, M.ϵ, τs)

end

Algorithm 12.4. The update func-
tion for the natural policy gradi-
ent, given policy π(θ, s), for an
MDP 𝒫 with initial state distribu-
tion b. The natural gradient with
respect to the parameter vector θ is
estimated from m rollouts to depth
d using the log policy gradients
∇logπ. The natural_update helper
method conducts an update ac-
cording to equation (12.12), given
an objective gradient ∇f(τ) and a
Fisher matrix F(τ) for a list of tra-
jectories.

During the line search phase, we no longer use a first-order approximation. In-
stead, we use an approximation derived from an equality involving the advantage
function5 5 A variation of this equality is

proven in lemma 6.1 of S.M.
Kakade and J. Langford, “Approx-
imately Optimal Approximate Re-
inforcement Learning,” in Interna-
tional Conference on Machine Learn-
ing (ICML), 2002.

U(θ′) = U(θ) + E
τ∼π

θ′

[

d

∑
k=1

Aθ(s
(k), a(k))

]

(12.14)

Anotherway towrite this is to use bγ,θ, which is the discounted visitation distribution
of state s under policy πθ, where

bγ,θ(s) ∝ P(s(1) = s) + γP(s(2) = s) + γ2P(s(3) = s) + · · · (12.15)

Using the discounted visitation distribution, the objective becomes

U(θ′) = U(θ) + E
s∼bγ,θ′

[

E
a∼π

θ′ (·|s)
[Aθ(s, a)]

]

(12.16)

We would like to pull our samples from our policy parameterized by θ instead
of θ′ so that we do not have to run more simulations during the line search. The
samples associated with the inner expectation can be replaced with samples from
our original policy so long as we appropriately weight the advantage:6 6 This weighting comes from impor-

tance sampling, which is reviewed
in appendix A.14.

U(θ′) = U(θ) + E
s∼bγ,θ′

[

E
a∼πθ(·|s)

[

πθ′(a | s)

πθ(a | s)
Aθ(s, a)

]]

(12.17)

The next step involves replacing the state distribution with bγ,θ. The quality of
the approximation degrades as θ′ gets further from θ, but it is hypothesized that
it is acceptable within the trust region. Since U(θ) does not depend on θ′, we can
drop it from the objective. We can also drop the state value function from the
advantage function, leaving us with the action value function. What remains is
referred to as the surrogate objective:

f (θ,θ′) = E
s∼bγ,θ

[

E
a∼πθ(·|s)

[

πθ′(a | s)

πθ(a | s)
Qθ(s, a)

]]

(12.18)

This equation can be estimated from the same set of trajectories that was used to
estimate the natural gradient update. We can estimate Qθ(s, a) using the reward-
to-go in the sampled trajectories.7

7 Algorithm 12.5 instead uses
∑ℓ=k r(ℓ)γℓ−1, which effectively
discounts the reward-to-go by
γk−1. This discount is needed to
weight each sample’s contribution
to match the discounted visitation
distribution. The surrogate
constraint is similarly discounted.

The surrogate constraint in the line search is given by

g(θ,θ′) = E
s∼bγ,θ

[DKL(πθ(· | s) || πθ′(· | s))] ≤ ǫ (12.19)

Line search involves iteratively evaluating our surrogate objective f and sur-
rogate constraint g for different points in the policy space. We begin with the θ′
obtained from the same process as the natural gradient update.We then iteratively
apply

θ′ ← θ+ α(θ′ − θ) (12.20)
until we have an improvement in our objective with f (θ,θ′) > f (θ,θ) and our
constraint is met with g(θ,θ′) ≤ ǫ. The step factor 0 < α < 1 shrinks the distance
between θ and θ′ at each iteration, with α typically set to 0.5.

Algorithm 12.5 provides an implementation of this approach. Figure 12.4
illustrates the relationship between the feasible regions associatedwith the natural
gradient and the line search. Figure 12.5 demonstrates the approach on a regulator
problem, and example 12.1 shows an update for a simple problem.

12.5 Clamped Surrogate Objective

We can avoid detrimental policy updates from overly optimistic estimates of
the trust region surrogate objective by clamping.8 The surrogate objective from 8 Clamping is a key idea in what is

known as proximal policy optimiza-
tion (PPO) as discussed by J. Schul-
man, F. Wolski, P. Dhariwal, A.
Radford, andO. Klimov, “Proximal
Policy Optimization Algorithms,”
2017. arXiv: 1707.06347v2.

equation (12.18), after exchanging the action value advantage, is

E
s∼bγ,θ

[

E
a∼πθ(·|s)

[

πθ′(a | s)

πθ(a | s)
Aθ(s, a)

]]

(12.21)

The probability ratio πθ′(a | s)/πθ(a | s) can be overly optimistic. A pessimistic
lower bound on the objective can significantly improve performance:

E
s∼bγ,θ

[

E
a∼πθ(·|s)

[

min

(

πθ′(a | s)

πθ(a | s)
Aθ(s, a), clamp

(

πθ′(a | s)

πθ(a | s)
, 1− ǫ, 1 + ǫ

)

Aθ(s, a)

)]]

(12.22)

where ǫ is a small positive value9 and clamp(x, a, b) forces x to be between a and 9 While this ǫ does not directly act
as a threshold on divergence, as it
did in previous algorithms, its role
is similar. A typical value is 0.2.

b. By definition, clamp(x, a, b) = min{max{x, a}, b}.
Clamping the probability ratio alone does not produce a lower bound; we must

also take the minimum of the clamped and original objectives. The lower bound
is shown in figure 12.6, together with the original and clamped objectives. The
end result of the lower bound is that the change in probability ratio is ignored
when it would cause the objective to improve significantly. Using the lower bound
thus prevents large, often detrimental, updates in these situations and removes
the need for the trust region surrogate constraint equation (12.19). Without the

struct TrustRegionUpdate
𝒫 # problem
b # initial state distribution
d # depth
m # number of samples
π # policy π(s)
p # policy likelihood p(θ, a, s)
∇logπ # log likelihood gradient
KL # KL divergence KL(θ, θ′, s)
ϵ # divergence bound
α # line search reduction factor (e.g., 0.5)

end

function surrogate_objective(M::TrustRegionUpdate, θ, θ′, τs)
d, p, γ = M.d, M.p, M.𝒫.γ
R(τ, j) = sum(r*γ^(k-1) for (k,(s,a,r)) in zip(j:d, τ[j:end]))
w(a,s) = p(θ′,a,s) / p(θ,a,s)
f(τ) = mean(w(a,s)*R(τ,k) for (k,(s,a,r)) in enumerate(τ))
return mean(f(τ) for τ in τs)

end

function surrogate_constraint(M::TrustRegionUpdate, θ, θ′, τs)
γ = M.𝒫.γ
KL(τ) = mean(M.KL(θ, θ′, s)*γ^(k-1) for (k,(s,a,r)) in enumerate(τ))
return mean(KL(τ) for τ in τs)

end

function linesearch(M::TrustRegionUpdate, f, g, θ, θ′)
fθ = f(θ)
while g(θ′) > M.ϵ || f(θ′) ≤ fθ

θ′ = θ + M.α*(θ′ - θ)
end
return θ′

end

function update(M::TrustRegionUpdate, θ)
𝒫, b, d, m, ∇logπ, π, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.π, M.𝒫.γ
πθ(s) = π(θ, s)
R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))
∇log(τ) = sum(∇logπ(θ, a, s) for (s,a) in τ)
∇U(τ) = ∇log(τ)*R(τ)
F(τ) = ∇log(τ)*∇log(τ)'
τs = [simulate(𝒫, rand(b), πθ, d) for i in 1:m]
θ′ = natural_update(θ, ∇U, F, M.ϵ, τs)
f(θ′) = surrogate_objective(M, θ, θ′, τs)
g(θ′) = surrogate_constraint(M, θ, θ′, τs)
return linesearch(M, f, g, θ, θ′)

end

Algorithm 12.5. The update pro-
cedure for trust region policy opti-
mization, which augments the nat-
ural gradient with a line search. It
generates m trajectories using pol-
icy π in problem 𝒫 with initial state
distribution b and depth d. To ob-
tain the starting point of the line
search, we need the gradient of the
log-probability of the policy gener-
ating a particular action from the
current state, which we denote as
∇logπ. For the surrogate objective,
we need the probability function
p, which gives the probability that
our policy generates a particular
action from the current state. For
the surrogate constraint, we need
the divergence between the action
distributions generated by πθ and
πθ′ . At each step of the line search,
we shrink the distance between the
considered point θ′ and θ while
maintaining the search direction.

Figure 12.4. Trust region policy op-
timization searches within the el-
liptical constraint generated by a
second-order approximation of the
Kullback-Leibler divergence. After
computing the natural policy gra-
dient ascent direction, a line search
is conducted to ensure that the up-
dated policy improves the policy
reward and adheres to the diver-
gence constraint. The line search
starts from the estimated maxi-
mum step size and reduces the step
size along the ascent direction until
a satisfactory point is found.

Figure 12.5. Trust region policy op-
timization applied to the simple
regulator problem with rollouts to
depth 10 with ǫ = 1 and c = 2. The
optimal policy parameterization is
shown in black.

Consider applying TRPO to the Gaussian policyN (θ1, θ2
2) from example 11.3

to the single-stateMDP from example 11.1 with γ = 1. Recall that the gradient
of the log policy likelihood is

∂

∂θ1
log πθ(a | s) =

a− θ1

θ2
2

∂

∂θ2
log πθ(a | s) =

(a− θ1)
2 − θ2

2

θ3
2

Suppose that we run two rollouts with θ = [0, 1] (this problem only has
one state):

τ1 = {(a = r = −0.532), (a = r = 0.597), (a = r = 1.947)}
τ2 = {(a = r = −0.263), (a = r = −2.212), (a = r = 2.364)}

The estimated Fisher information matrix is

Fθ =
1

2

(

∇ log p(τ(1))∇ log p(τ(1))⊤ +∇ log p(τ(2))∇ log p(τ(2))⊤
)

=
1

2

([

4.048 2.878

2.878 2.046

]

+

[

0.012 − 0.838

−0.838 57.012

])

=

[

2.030 1.020

1.019 29.529

]

The objective function gradient is [2.030, 1.020]. The resulting descent di-
rection u is [1, 0]. Setting ǫ = 0.1, we compute our updated parameterization
vector and obtain θ′ = [0.314, 1].

The surrogate objective function value at θ is 1.485. Line search begins at
θ′, where the surrogate objective function value is 2.110 and the constraint
yields 0.049. This satisfies our constraint (as 0.049 < ǫ), so we return the
new parameterization.

Example 12.1. An example of one
iteration of trust region policy op-
timization.

constraint, we can also eliminate line search and use standard gradient ascent
methods.

0 1− ǫ 1 1 + ǫ

probability ratio

jec

tiv

A > 0

0 1− ǫ 1 1 + ǫ

A < 0

original objective clamped objective lower-bound objective

Figure 12.6. A visualization of
the lower-bound objective for posi-
tive and negative advantages com-
pared to the original objective
and the clamped objective. The
black point shows the baseline
around which the optimization is
performed, πθ′ (a | s)/πθ(a | s) =
1. The three line plots in each axis
are vertically separated for clarity.

The gradient of the unclamped objective equation (12.21) with action values is

∇θ′ f (θ,θ′) = E
s∼bγ,θ

[

E
a∼πθ(·|s)

[∇θ′πθ′(a | s)

πθ(a | s)
Qθ(s, a)

]]

(12.23)

where Qθ(s, a) can be estimated from reward-to-go. The gradient of the lower-
bound objective equation (12.22) (with clamping), is the same, except there is no
contribution from experience tuples for which the objective is actively clamped.
That is, if either the reward-to-go is positive and the probability ratio is greater
than 1 + ǫ, or if the reward-to-go is negative and the probability ratio is less than
1− ǫ, the gradient contribution is zero.

Like TRPO, the gradient can be computed for a parameterization θ′ from
experience generated from θ. Hence, several gradient updates can be run in
a row using the same set of sampled trajectories. Algorithm 12.6 provides an
implementation of this.

The clamped surrogate objective is compared to several other surrogate ob-
jectives in figure 12.7, which includes a line plot for the effective objective for
TRPO:

E
s∼bγ,θ

a∼πθ(·|s)

[

πθ′(a | s)

πθ(a | s)
Aθ(s, a)− βDKL(πθ(· | s) || πθ′(· | s))

]

(12.24)

struct ClampedSurrogateUpdate
𝒫 # problem
b # initial state distribution
d # depth
m # number of trajectories
π # policy
p # policy likelihood
∇π # policy likelihood gradient
ϵ # divergence bound
α # step size
k_max # number of iterations per update

end

function clamped_gradient(M::ClampedSurrogateUpdate, θ, θ′, τs)
d, p, ∇π, ϵ, γ = M.d, M.p, M.∇π, M.ϵ, M.𝒫.γ
R(τ, j) = sum(r*γ^(k-1) for (k,(s,a,r)) in zip(j:d, τ[j:end]))
∇f(a,s,r_togo) = begin

P = p(θ, a,s)
w = p(θ′,a,s) / P
if (r_togo > 0 && w > 1+ϵ) || (r_togo < 0 && w < 1-ϵ)

return zeros(length(θ))
end
return ∇π(θ′, a, s) * r_togo / P

end
∇f(τ) = mean(∇f(a,s,R(τ,k)) for (k,(s,a,r)) in enumerate(τ))
return mean(∇f(τ) for τ in τs)

end

function update(M::ClampedSurrogateUpdate, θ)
𝒫, b, d, m, π, α, k_max= M.𝒫, M.b, M.d, M.m, M.π, M.α, M.k_max
πθ(s) = π(θ, s)
τs = [simulate(𝒫, rand(b), πθ, d) for i in 1:m]
θ′ = copy(θ)
for k in 1:k_max

θ′ += α*clamped_gradient(M, θ, θ′, τs)
end
return θ′

end

Algorithm 12.6. An implementa-
tion of clamped surrogate policy
optimization, which returns a new
policy parameterization for policy
π(s) of an MDP 𝒫 with initial state
distribution b. This implementa-
tion samples m trajectories to depth
d, and then uses them to estimate
the policy gradient in k_max subse-
quent updates. The policy gradient
using the clamped objective is con-
structed using the policy gradients
∇p with clamping parameter ϵ.

which is the trust region policy objective where the constraint is implemented
as a penalty for some coefficient β. TRPO typically uses a hard constraint rather
than a penalty because it is difficult to choose a value of β that performs well
within a single problem, let alone across multiple problems.

−0.4 −0.2 0 0.2 0.4 0.6 0.8 1 1.2 1.4 1.6 1.8 2
−2

0

2

4

linear interpolation factor

surrogate objective
surrogate constraint
TRPO effective objective
clamped surrogate objective

Figure 12.7. A comparison of surro-
gate objectives related to clamped
surrogate policy optimization us-
ing the linear quadratic regulator
problem. The x-axis shows surro-
gate objectives as we travel from
θ at 0 toward θ′, given a natural
policy update at 1. The surrogate
objectives were centered at 0 by
subtracting the surrogate objective
function value for θ. We see that
the clamped surrogate objective be-
haves very similarly to the effective
TRPO objective without needing a
constraint. Note that ǫ and β can
be adjusted for both algorithms,
which would affect where the max-
imum is in each case.

12.6 Summary

• The gradient ascent algorithm can use the gradient estimates obtained from the
methods discussed in the previous chapter to iteratively improve our policy.

• Gradient ascent can be made more robust by scaling, clipping, or forcing the
size of the improvement steps to be uniform.

• The natural gradient approach uses a first-order approximation of the objective
function with a constraint on the divergence between the trajectory distribution
at each step, approximated using an estimate of the Fisher information matrix.

• Trust region policy optimization involves augmenting the natural gradient
method with a line search to further improve the policy without additional
trajectory simulations.

• We can use a pessimistic lower bound of the TRPO objective to obtain a clamped
surrogate objective that performs similarly without the need for line search.

12.7 Exercises
Exercise 12.1. TRPO starts its line search from a new parameterization given by a natural
policy gradient update. However, TRPO conducts the line search using a different objec-
tive than the natural policy gradient. Show that the gradient of the surrogate objective
equation (12.18) used in TRPO is actually the same as the reward-to-go policy gradient
equation (11.26).

Solution: The gradient of TRPO’s surrogate objective is

∇θ′UTRPO = E
s∼bγ,θ

[

E
a∼πθ(·|s)

[∇θ′πθ′ (a | s)

πθ(a | s)
Qθ(s, a)

]]

When conducting the initial natural policy gradient update, the search direction is
evaluated at θ′ = θ. Furthermore, the action value is approximated with the reward-to-go:

∇θ′UTRPO = E
s∼bγ,θ

[

E
a∼πθ(·|s)

[∇θπθ(a | s)

πθ(a | s)
rto-go

]]

Recall that the derivative of log f (x) is f ′(x)/ f (x). It thus follows that

∇θ′UTRPO = E
s∼bγ,θ

[

E
a∼πθ(·|s)

[

∇θ log πθ(a | s)rto-go
]

]

which takes the same form as the reward-to-go policy gradient equation (11.26).

Exercise 12.2. Perform the calculations of example 12.1. First, compute the inverse of the
Fisher information matrix F−1

θ
, compute u, and compute the updated parameters θ′.

Solution: We start by computing the inverse of the Fisher information matrix:

F−1
θ
≈ 1

0.341(29.529)− 0.332(0.332)

[

29.529 −0.332

−0.332 0.341

]

≈
[

0.501 −0.017

−0.017 0.034

]

Now, we update u as follows:

u = F−1
θ
∇U(θ) ≈

[

0.501 −0.017

−0.017 0.034

] [

2.030

1.020

]

≈
[

1

0

]

Finally, we estimate the updated parameters θ:

θ′ = θ+ u

√

2ǫ

∇U(θ)⊤u

≈
[

0

1

]

+

[

1

0

]√

√

√

√

√

√

2(0.1)
[

2.030 1.020
]

[

1

0

]

≈
[

0

1

]

+

[

1

0

]

√

0.2

2.030

≈
[

0.314

1

]

Exercise 12.3. Suppose we have the parameterized policies πθ and πθ′ given in the
following table:

a1 a2 a3 a4

πθ(a | s1) 0.1 0.2 0.3 0.4

πθ′ (a | s1) 0.4 0.3 0.2 0.1

πθ(a | s2) 0.1 0.1 0.6 0.2

πθ′ (a | s2) 0.1 0.1 0.5 0.3

Given thatwe sample the followingfive states, s1, s2, s1, s1, s2, approximateEs [DKL(πθ(· | s) || πθ′ (· | s))]

using the definition
DKL(P || Q) = ∑

x
P(x) log

P(x)

Q(x)

Solution: First, we compute the KL divergence for a state sample s1:

DKL(πθ(· | s1) || πθ′ (· | s1)) = 0.1 log
(

0.1
0.4

)

+ 0.2 log
(

0.2
0.3

)

+ 0.3 log
(

0.3
0.3

)

+ 0.4 log
(

0.4
0.1

)

≈ 0.456

Now, we compute the KL divergence for a state sample s2:

DKL(πθ(· | s2) || πθ′ (· | s2)) = 0.1 log
(

0.1
0.1

)

+ 0.1 log
(

0.1
0.1

)

+ 0.6 log
(

0.6
0.5

)

+ 0.2 log
(

0.2
0.3

)

≈ 0.0283

Finally, we compute the approximation of the expectation, which is the average KL diver-
gence of the parameterized policies over the n state samples:

Es[DKL(πθ(· | s) || πθ′ (· | s))] ≈ 1

n

n

∑
i=1

DKL
(

πθ(· | s(i))
∣

∣

∣

∣

∣

∣
πθ′ (· | s(i))

)

≈ 1

5
(0.456 + 0.0283 + 0.456 + 0.456 + 0.0283)

≈ 0.285

